

## 2. ANALIZA STAŁOPRĄDOWA

W poprzednim rozdziale opisany został format danych wejściowych dla programu PSpice. Struktura obwodu określona jest przez deklarację podstawowych elementów elektronicznych. Wymienione zostały ograniczenia, jakie program PSpice narzuca na strukturę analizowanego obwodu. Wyjaśniono także ich przyczyny. Teraz należy przedstawić możliwości analizy obwodu jakie oferuje program PSpice.

### 2.1. Statyczny punkt pracy układu

Większość kursów elektrotechniki zaczyna się od metod analizy obwodu prądu stałego. Jest to problem obliczenia statycznego punktu pracy układu. W przypadku, gdy wszystkie elementy obwodu są liniowe problem ten ma zawsze rozwiązanie analityczne<sup>1</sup>. Jeśli jednak w obwodzie pojawiają się elementy o nieliniowej charakterystyce problem ten staje się z reguły bardzo skomplikowany. Zdarza się, że stosowane do obliczeń metody numeryczne zawodzą. Dotyczy to także programu PSpice mimo, że algorytmy zastosowane przez autorów programu uważane są za bardzo dobre [31].

W poprzednim rozdziale stwierdzono, że jeżeli zbiór danych wejściowych dla programu PSpice zawiera tylko opis struktury obwodu to automatycznie znaleziony zostanie statyczny punkt pracy układu. Wyniki analizy umieszczone zostaną w zbiorze danych wyjściowych. Czasami zdarza się jednak, że konieczne jest bezpośrednie polecenie wykonania tej analizy. Służy do tego instrukcja .OP (ang. operating point — punkt pracy):

#### **.OP**

Stacyjny punkt pracy obliczany jest przy założeniu, że każda cewka stanowi zwarcie natomiast każdy kondensator stanowi rozwarcie. Instrukcja .OP *nie musi* być stosowana (stacyjny punkt pracy obliczany jest automatycznie) w następujących przypadkach:

- ☐ Przed instrukcją służącą do obliczania stanu nieustalonego.
- ☐ Przed instrukcją służącą do obliczania małosygnałowych transmitancji stałoprądowych.

---

<sup>1</sup>Z wyjątkiem obwodów patologicznych takich jak np. równolegle połączone dwie idealne siły prądomotoryczne, każda o innej wartości.

- ☐ Przed instrukcją służącą do uruchomienia małosygnałowej analizy zmiennoprądowej.

## 2.2. Charakterystyki statyczne

Program PSpice pozwala na wykonanie ciągu analiz polegających na znajdowaniu statycznego punktu pracy układu przy zmieniających się parametrach obwodu. Służy do tego instrukcja `.DC` (ang. direct current — prąd stały). Jej format jest następujący:

```
.DC [LIN] _par _start _stop _krok [_par2 _start2 _stop2 _krok2]
.DC [OCT][DEC] _par _start _stop _li [_par _start2 _stop2 _li2]
.DC _par <_lista> [_par2 <_lista2>]
```

Przykłady:

```
.DC VIN -0.25 0.25 0.05
.DC LIN 12 I2 5mA -2mA 0.1mA
.DC VCE 0V 10V .5V IB 0mA 1mA 50UA
.DC RES RMOD(R) 0.9K 1.1K 1
.DC DEC NPN QFAST(IS) 1E-18 1E-14 5
.DC TEMP LIST 0 20 27 50 80 100 -20
```

Podczas analizy zmieniana jest wartość parametru o nazwie `_par`. Pole `_par` może zawierać:

- ☐ Nazwę niezależnego źródła napięcia lub prądu. Zmieniana jest wówczas wydajność źródła.
  - ☐ Nazwę modelu poprzedzoną typem modelu. Zmieniana jest wartość parametru modelu podanego w nawiasie, tuż za nazwą modelu (bez spacji) — patrz przykład czwarty i piąty powyżej.
  - ☐ Słowo kluczowe **TEMP**. Zmieniana jest wówczas temperatura analizowanego układu.
- Zmiany wymienionych wielkości mogą zachodzić liniowo, logarytmicznie lub wg listy wartości. Dozwolone są następujące typy zmian:

- LIN** Wartość parametru zmienia się od wartości `_start` do wartości `_stop` liniowo z krokiem `_krok`. Słowo kluczowe **LIN** może zostać pominięte.
- OCT** Wartość parametru zmienia się logarytmicznie co oktawę od wartości `_start` do wartości `_stop`, przy czym liczba punktów w każdej oktawie wynosi `_li`.
- DEC** Wartość parametru zmienia się logarytmicznie co dekadę od wartości `_start` do wartości `_stop`, przy czym liczba punktów w każdej dekadzie wynosi `_li`.
- LIST** Wartość parametru zmienia się wg listy. Brak jest wartości początkowej i końcowej natomiast liczby, które następują po słowie kluczowym **LIST** (pole `<_lista>`), stanowią kolejne wartości parametru.

W polu `_par2` można podać drugi parametr, który będzie zmieniany podczas analizy. W tym wypadku dla każdej wartości drugiego parametru `_par2` wykonany zostanie ciąg analiz, w którym wartość pierwszego parametru `_par` zmienia się w pełnym zakresie.

Przykład:

Instrukcja .DC posłużyć może dla obliczenia charakterystyki statycznej tranzystora MOS. Dla tego typu elementu istnieje w programie PSpice wbudowany model matematyczny. W celu wykonania obliczeń tworzymy zbiór wejściowy pokazany poniżej.

```
CHARAKTERYSTYKA TRANZYSTORA MOS
VDS 3 0                                ;źródło napięcia zasilające dren
VGS 2 0                                ;źródło napięcia zasilające bramkę
M1 1 2 0 0 MODEL_MOS                  ;deklaracja tranzystora MOS w strukturze obwodu
*                                     nazwa modelu tranzystora
*                                     podłoże
*                                     źródło
*                                     bramka
*                                     dren
*deklaracja modelu matematycznego tranzystora MOS
.MODEL MODEL_MOS NMOS VTO=-2V NSUB=1.0E15 UO=550 L=4U W=6U
*                                     napięcie progowe
*                                     domieszkowanie
*                                     podłoża
*                                     ruchliwość nośników
*                                     długość i szerokość kanału
VIDS 3 1                               ;źródło służące do pomiaru prądu drenu
.DC VDS 0 10 0.1 VGS 0 5 1             ;instrukcja analizy stałoprądowej
.PROBE I(VIDS)                          ;na wy. graficzne przekazane zostaną wartości prądu źródła VIDS
.END                                    ;koniec danych wejściowych
```


Powyższe dane stanowią opis obwodu przedstawionego na Rys. 8. W czwartej linii znajduje się deklaracja tranzystora MOS. Nazwa tranzystora MOS zaczyna się zawsze od litery „M”. Dalej podane są numery węzłów, w które wpięte są dren, bramka, źródło i podłoże tranzystora. Na końcu linii deklaracji znajduje się nazwa modelu tranzystora. W naszym przypadku jest to MODEL\_MOS. Model ten zadeklarowany jest niżej za pomocą deklaracji .MODEL służącej do deklarowania modeli przyrządów. W deklaracji tej wymieniona jest nazwa modelu i jego typ. W naszym przypadku typ modelu to NMOS, co oznacza model tranzystora MOS z kanałem typu N. W linii deklaracji modelu podaje się wartości parametrów modelu<sup>2</sup>. W danych do przykładu podane zostały tylko niektóre z nich:

VTO napięcie progowe; wymiar [V].

NSUB koncentracja atomów domieszek w podłożu; wymiar [ $\text{cm}^{-3}$ ].

UO ruchliwość nośników tuż przy powierzchni półprzewodnika; wymiar [ $\text{cm}^2/(\text{V} \cdot \text{s})$ ].

L długość kanału tranzystora; wymiar [m].



**Rys.8.** Obwód służący do obliczenia charakterystyki statycznej tranzystora MOS.

<sup>2</sup>Model matematyczny tranzystora MOS wbudowany w program PSpice, znaleźć można w rozdziale 6.

W szerokość kanału tranzystora; wymiar [m].

W języku symulacyjnym programu PSpice celowo rozdzielono deklarację elementu półprzewodnikowego w strukturze obwodu i deklarację modelu elementu półprzewodnikowego. Podczas wytwarzania układów scalonych<sup>3</sup> elementy tego samego typu np. tranzystory MOS powstają w jednym procesie technologicznym. Posiadają zatem te same parametry elektryczne<sup>4</sup>. Co więcej, zmiany parametrów wraz z temperaturą czy też w wyniku błędów powstałych w procesie technologicznym są silnie skorelowane. Celowe jest zatem opisywanie wszystkich przyrządów tego samego typu tym samym modelem matematycznym. Po deklaracji tranzystora MOS następuje deklaracja SEM o nazwie VIDS i wartości równej zero. Służy ona do pomiaru prądu płynącego przez dren tranzystora. Można w tym celu użyć siły elektromotorycznej VDS pamiętając, że płynący przez nią prąd ma wartość ujemną. Następna linia zawiera instrukcję obliczania charakterystyk statycznych (.DC). Dla każdej wartości napięcia między bramką a źródłem VGS, zmieniającej się od wartości 0[V] do wartości 5[V] co 1[V], obliczany jest prąd drenu w funkcji zmian napięcia dren–źródło VDS. Napięcie VDS zmienia się przy tym od 0[V] do 10[V] z krokiem 100[mV]. Za pomocą instrukcji .PROBE wyniki obliczeń przekazywane są do specjalnego programu graficznego o nazwie Probe<sup>5</sup>. Program ten jest rozprowadzany przez firmę MicroSim wraz z programem PSpice. Probe służy do graficznej ilustracji wyników obliczeń przeprowadzonych za pomocą programu PSpice. Parametrami instrukcji .PROBE są wielkości, które użytkownik chce otrzymać w postaci wykresu. W naszym przypadku jest to prąd drenu tranzystora czyli prąd płynący przez źródło napięcia o nazwie VIDS. Prąd ten oznaczany jest jako I(VIDS). Dane kończą się instrukcją .END (ang. end — koniec) — koniec danych wejściowych.

Wyniki obliczeń wykonanych przez program PSpice przedstawione są na Rys. 9. Jest to rodzina krzywych przedstawiająca zależność prądu drenu od napięcia dren–źródło VDS dla tranzystora MOS. Parametrem rodziny jest napięcie bramka–źródło VGS. Przedstawiony wykres uzyskany został za pomocą programu Probe. Rys. 9 jest bardzo zbliżony do tego co wyświetlane jest na ekranie monitora komputerowego.

### 2.3. Zbieżność obliczeń


Tak jak już stwierdzono problem znalezienia statycznego punktu pracy nieliniowego

---

<sup>3</sup>SPICE2 - pierwowzór programu PSpice przeznaczony był głównie do analizy układów scalonych.

<sup>4</sup>Program PSpice posiada mechanizmy pozwalające na modyfikację parametrów, które zmieniają się wraz z wymiarami geometrycznymi przyrządu.

<sup>5</sup>Więcej informacji na temat programu Probe czytelnik znajdzie w dodatkach C i D.



**Rys.9.** Zależność prądu drenu od napięcia dren-źródło w tranzystorze MOS. Parametrem rodziny krzywych jest napięcie bramka-źródło.

układu elektronicznego jest zwykle trudnym problemem numerycznym. Zdaża się, że obliczenia wykonywane przez program PSpice według algorytmu Newton-a Raphson-a [5],[28],[30] są niezbieżne. Dzieje się tak, wtedy gdy początkowe wartości uogólnionych potencjałów węzłowych opisujących układ są zbyt dalekie od właściwego rozwiązania [5].

Przykład:


Dany jest układ bramki TTL (negator) przedstawiony na Rys. 10. Wejście bramki sterowane jest przez źródło V2 o wartości 1.58[V]. Należy obliczyć wszystkie potencjały węzłowe w tym układzie.

Dane dla programu PSpice przedstawione są poniżej. Zwróćmy uwagę, że:

- ☐ Linia deklaracji tranzystora bipolarnego zaczyna się od jego nazwy. Nazwa tranzystora zaczyna się na literę „Q”. Dalej podane są numery węzłów, do których dołączone są odpowiednio: kolektor, baza i emiter tranzystora. Na końcu linii znajduje się nazwa modelu tranzystora.
- ☐ Model tranzystora bipolarnego zdefiniowany jest za pomocą poznanej już wcześniej deklaracji .MODEL. Nazwa modelu to TR natomiast typ NPN. Oznacza to tranzystor bipolarny typu n-p-n. W linii deklaracji tranzystora wyszczególnione są parametry modelu. W naszym wypadku jedynym parametrem, którego wartość zadeklarowano

jest wzmacnienie prądowe tranzystora dla pracy normalnej BF równe 100.

- Nazwa diody półprzewodnikowej zaczyna się od litery D. Po nazwie deklaruje się numery węzłów, do których dołączono odpowiednio anodę i katodę diody. Dalej następuje nazwa modelu diody.



Rys.10. Bramka TTL.

- Model diody określony jest za pomocą deklaracji `.MODEL`. Typ modelu D oznacza diodę półprzewodnikową. W linii deklaracji nie ma deklaracji wartości jakichkolwiek parametrów (umieszcza się je na końcu linii). Wobec tego przyjęte zostaną wartości domyślne<sup>6</sup>.
- W drugiej linii danych wejściowych znajduje się instrukcja `.OPTIONS`<sup>7</sup>. Z jej pomocą można zmienić wiele parametrów określających sposób, w jaki program PSpice dokonuje obliczeń. Opcja `NOPAGE`, umieszczona w przykładowych danych powoduje, że zbiór wyjściowy nie będzie dzielony na strony.

```
BRAMKA TTL
.OPTIONS NOPAGE                ;dane wyjściowe bez podziału na strony
.MODEL TR NPN BF=100           ;deklaracja modelu mat. tranzystora bipolarnego o wzmacnieniu
prądowym =100
.MODEL DIO D                   ;deklaracja modelu diody - parametry domyślne
V1 3 0 5V                      ;źródło zasilania
V2 10 0 1.58V                  ;źródło sterujące
RW 1 10 100                    ;oporność źródła sterującego
*****
* oporniki bramki *
*****
R1 3 9 4K
R2 3 4 1.6K
R3 5 0 1K
R4 3 6 100
*****
* tranzystory bramki *
*****
Q1 2 9 1 TR
Q2 4 2 5 TR
Q3 6 4 7 TR
Q4 8 5 0 TR
*****
```

<sup>6</sup>Dokładne omówienie modelu diody półprzewodnikowej wbudowanego w program PSpice znajduje się w rozdziale 6.

<sup>7</sup>Pełna lista opcji instrukcji `.OPTIONS` znajduje się w dodatku A.

```

* diody bramki *
*****
D1 7 8 DIO
D2 0 1 DIO
.OP                      ;oblicz punkt pracy
.END                     ;koniec danych


```

Po uruchomieniu programu PSpice okazuje się jednak, że symulator nie jest w stanie obliczyć statycznego punktu pracy układu. W zbiorze wyjściowym znajdujemy informację, że obliczenia zostały przerwane z powodu braku zbieżności.

### 2.3.1. Algorytm Newton–a Raphson–a

Aby zrozumieć istotę trudności na jakie natrafiliśmy trzeba zapoznać się z zastosowanym algorytmem obliczeń. Problem obliczania statycznego punktu pracy nieliniowego układu elektrycznego, takiego jak np. bramka TTL, sprowadza się do rozwiązywania nieliniowego układu równań algebraicznych. W najprostszym przypadku jest to jedno równanie w postaci:

$$f(x)=0 \quad (15)$$



**Rys.11.** Przykładowy wykres funkcji nieliniowej jednej zmiennej.

Funkcja  $f(x)$  jest nieliniowa. Oznacza to, że jej wykres może wyglądać tak jak na Rys. 11. W programie PSpice do rozwiązywania równań w rodzaju (15) zastosowano algorytm Newtona–a Raphsona. Jest to algorytm iteracyjny. Załóżmy, że znamy przybliżoną wartość rozwiązania równania (15). Oznaczmy ją przez  $x_0$ . Rozwiązanie dokładne oznaczmy natomiast przez  $x^*$ . Aby znaleźć wartość  $x_1$  bardziej zbliżoną do rozwiązania dokładnego rozwińmy funkcję  $f(x)$  w szereg Taylora wokół punktu  $x_0$  i podstawmy do równania (15):

$$f(x_-) = f(x_0) + \frac{f'(x_0)}{1!} \cdot (x_- - x_0) + \frac{f''(x_0)}{2!} \cdot (x_- - x_0)^2 + \dots = 0 \quad (16)$$

Jeżeli odrzucimy wszystkie wyrazy rozwinięcia rzędu wyższego niż 1 otrzymamy równanie przybliżone:

$$f(x_0) + f'(x_0) \cdot (x_- - x_0) \approx 0 \quad (17)$$

Stąd:

$$x_- \approx x_1 = x_0 - \frac{f(x_0)}{f'(x_0)} \quad (18)$$

Aby otrzymać lepsze przybliżenie niż  $x_1$  stosujemy opisaną procedurę, lecz tym razem w stosunku do przybliżenia  $x_1$ . Postępujemy w ten sposób tak długo aż uzyskane przybliżenie jest wystarczająco dokładne<sup>8</sup>. W ten sposób otrzymujemy formułę rekurencyjną pozwalającą na rozwiązanie równania (15) z dowolną dokładnością:

$$x_{n+1} = x_n - \frac{f(x_n)}{f'(x_n)} \quad (19)$$

Rozwinięcie funkcji  $f(x)$  w szereg Taylora wokół punktu  $x_0$  i odrzucenie wyrazów rzędu wyższego niż 1 można interpretować w sposób następujący:

□ Funkcję  $f(x)$  zastępujemy funkcją liniową  $w(x)$ :

$$w(x) = f(x_0) + f'(x_0) \cdot (x - x_0) \quad (20)$$

Funkcja  $w(x)$  przyjmuje w punkcie  $x_0$  tę samą wartość co funkcja  $f(x)$ , a jednocześnie pochodna funkcji  $w(x)$  w punkcie  $x_0$  posiada tę samą wartość co pochodna funkcji  $f(x)$ :

$$w(x_0) = f(x_0) \quad ; \quad w'(x_0) = f'(x_0) \quad (21)$$

□ Przybliżone rozwiązanie  $x_1$  równania (15) jest dokładnym rozwiązaniem równania:

$$w(x) = 0 \quad (22)$$

Na Rys. 12 widać, że dla przykładowej funkcji  $f(x)$  i wartości  $x_0$  zbliżonej do  $x^*=0$  wartość  $x_1$  jest bliższa rozwiązania dokładnego niż  $x_0$ . Jeżeli jednak wartość początkowa rozwiązania  $x_0$  jest bardziej odległa niż to pokazano na Rys. 12 to przybliżone rozwiązanie  $x_1$  może być gorsze niż  $x_0$ . Ilustruje to Rys. 13. W tym przypadku każdy kolejny krok iteracji oddala nas od rozwiązania. Mówimy, że obliczenia są niebezpieczne.


---

<sup>8</sup>Osobnym problemem jest ściśle określenie co to znaczy "wystarczająco dokładne przybliżenie".



Z taką właśnie sytuacją spotkaliśmy się obliczając statyczny punkt pracy bramki TTL. Algorytm zastosowany w programie PSpice zakłada, że początkowe wartości uogólnionych potencjałów węzłowych wynoszą zero. Wartość ta jest jednak zbyt odległa od rozwiązania dokładnego. Stąd brak zbieżności obliczeń.

Zwróćmy uwagę, że w przypadku równania (15) nieliniową funkcję  $f(x)$  zastąpiono funkcją liniową  $w(x)$  daną wzorem (20). Dla obwodu odpowiada to zastąpieniu obwodu nieliniowego pewnym obwodem liniowym. Rozwiązanie równania (22) odpowiada natomiast znalezieniu potencjałów węzłowych tegoż obwodu liniowego. Zatem algorytm pozwalający na ułożenie i rozwiązanie równań obwodu liniowego jest kluczowym algorytmem symulatora układów elektronicznych.




**Rys.12.** Algorytm Newton-a Raphson-a znajdowania miejsca zerowego funkcji jest zbieżny.

### 2.3.2. Deklaracja

#### .NODESET

W rozważanym przykładzie aby uzyskać zbieżność obliczeń można użyć deklaracji .NODESET (ang. *node* — węzeł; *set* — ustal), która pozwala na rozpoczęcie iteracji od wartości bliskich właściwemu punktowi pracy układu. Składnia deklaracji .NODESET jest następująca:



**Rys.13.** Algorytm Newton-a Raphson-a znajdowania miejsca zerowego funkcji jest niezbieżny.

**.NODESET V(numer\_w1)=\_wartość1 [ V(numer\_w2)=\_wartość2 ...]**

Przykład:

.NODESET V(2)=3.4 V(102)=0 V(3)=-1V

Deklaracja .NODESET powoduje, że najpierw obliczane są wstępne wartości potencjałów

węzłowych układu. Podczas tych obliczeń potencjały w węzłach o numerach **numer\_w1**, **numer\_w2**, ... mają wartość stałą i są równe wartościom podanym w polach **\_wartość1**, **\_wartość2**, ... . Następnie rozpoczyna się obliczanie ostatecznych wartości potencjałów węzłowych. Punktem wyjścia do iteracji są obliczone wcześniej „wstępne” wartości potencjałów węzłowych. Instrukcja **.NODESET** jest zwykle wykorzystywana w celu:

- ☐ Uzyskania zbieżności obliczeń podczas obliczania statycznego punktu pracy układu.
- ☐ Wybrania do dalszych obliczeń jednego ze stanów stabilnych podczas analizy układu bistabilnego.

W naszym przypadku, dla osiągnięcia zbieżności obliczeń, należy zbiór danych wyjściowych uzupełnić o następującą linię:

**.NODESET V(2)=1.6 V(4)=0.87 V(5)=0.81**

Zbiór danych wyjściowych tworzony przez program PSpice po zakończeniu obliczeń przedstawiony jest poniżej.

```
***** 09/28/92 ***** Evaluation PSpice (Jan. 1988) ***** 14:19:03 *****
BRAMKA TTL
****      CIRCUIT DESCRIPTION
*****

.OPTIONS NOPAGE                ;dane wyjściowe bez podziału na strony
.MODEL TR NPN BF=100           ;deklaracja modelu mat. tranzystora bipolarnego o wzmacnieniu
prądowym =100
.MODEL DIO D                   ;deklaracja modelu diody - parametry domyślne
V1 3 0 5V                     ;źródło zasilania
V2 10 0 1.58V                  ;źródło sterujące
RW 1 10 100                    ;oporność źródła sterującego
*****
* oporniki bramki *
*****
R1 3 9 4K
R2 3 4 1.6K
R3 5 0 1K
R4 3 6 100
*****
* tranzystory bramki *
*****
Q1 2 9 1 TR
Q2 4 2 5 TR
Q3 6 4 7 TR
Q4 8 5 0 TR
*****
* diody bramki *
*****
D1 7 8 DIO
D2 0 1 DIO
.NODESET V(2)=1.6V V(4)=0.87V V(5)=0.81V
.OP                             ;oblicz punkt pracy
.END                             ;koniec danych

****      Diode MODEL PARAMETERS
          DIO
          IS 10.000000E-15

****      BJT MODEL PARAMETERS
          TR
          NPN
          IS 100.000000E-18
          BF 100
          NF 1
          BR 1
          NR 1
```

---

```

****      SMALL SIGNAL BIAS SOLUTION      TEMPERATURE = 27.000 DEG C

NODE      VOLTAGE      NODE      VOLTAGE      NODE      VOLTAGE      NODE      VOLTAGE
(   1)      1.6074 (   2)      1.6163 (   3)      5.0000 (   4)      .8694
(   5)      .8114 (   6)      5.0000 (   7)      .3843 (   8)      .0179
(   9)      2.3793 (  10)      1.5800

VOLTAGE SOURCE CURRENTS
NAME             CURRENT
V1              -3.237E-03
V2              2.745E-04

TOTAL POWER DISSIPATION 1.58E-02 WATTS

****      OPERATING POINT INFORMATION      TEMPERATURE = 27.000 DEG C

****      DIODES

NAME      D1      D2
MODEL     DIO     DIO
ID        1.42E-08 -1.62E-12
VD        3.66E-01 -1.61E+00
REQ       1.83E+06 1.00E+12
CAP       0.00E+00 0.00E+00

****      BIPOLAR JUNCTION TRANSISTORS

NAME      Q1      Q2      Q3      Q4
MODEL     TR      TR      TR      TR
IB        6.55E-04 3.81E-04 1.36E-10 2.15E-03
IC       -3.81E-04 2.58E-03 1.40E-08 1.42E-08
VBE      7.72E-01 8.05E-01 4.85E-01 8.11E-01
VBC      7.63E-01 7.47E-01 -4.13E+00 7.94E-01
VCE      8.90E-03 5.80E-02 4.62E+00 1.79E-02
BETADC   -5.81E-01 6.78E+00 1.03E+02 6.59E-06
GM       1.03E-02 1.13E-01 5.42E-07 8.15E-02
RPI      2.84E+03 7.89E+02 1.84E+08 6.13E+02
RX       0.00E+00 0.00E+00 0.00E+00 0.00E+00
RO       4.00E+01 7.43E+01 1.00E+12 1.23E+01
CBE      0.00E+00 0.00E+00 0.00E+00 0.00E+00
CBC      0.00E+00 0.00E+00 0.00E+00 0.00E+00
CBX      0.00E+00 0.00E+00 0.00E+00 0.00E+00
CJS      0.00E+00 0.00E+00 0.00E+00 0.00E+00
BETAAC   2.91E+01 8.94E+01 1.00E+02 5.00E+01
FT       1.63E+17 1.80E+18 8.63E+12 1.30E+18

JOB CONCLUDED

TOTAL JOB TIME          2.20

```

---

W zbiorze wyjściowym znajdujemy kolejno:

- ☐ Powtórzenie danych przekazanych do przetwarzania w zbiorze wejściowym.
- ☐ Parametry modelu diody — jedynym różnym od zera parametrem jest prąd nasycenia IS, który wynosi 0.01pA.
- ☐ Parametry modelu tranzystora:
  - IS    prąd nasycenia złączy;
  - BF    wzmacnienie prądowe przy pracy normalnej;
  - NF    współczynnik emisji dla pracy normalnej
  - BR    wzmacnienie prądowe przy pracy inwersyjnej;
  - NR    współczynnik emisji dla pracy inwersyjnej.
- ☐ Potencjały węzłowe.

- ☐ Prądy płynące przez źródła napięcia.
- ☐ Całkowitą moc rozpraszaną przez układ.
- ☐ Informacje o punkcie pracy przyrządów półprzewodnikowych w układzie. W przypadku diody są to:

ID     prąd płynący przez diodę;  
VD     napięcie panujące na diodzie;  
REQ   rezystancja dynamiczna diody REQ;  
CAP   całkowita pojemność diody.

W przypadku tranzystorów są to:

IB        prąd bazy;  
IC        prąd kolektora;  
VBE      napięcie baza–emiter;  
VBC      napięcie baza–kolektor;  
VCE      napięcie kolektor–emiter;  
BETADC   stałoprądowy współczynnik wzmocnienia prądowego;  
RX        rezystancja obszaru bazy;  
CBE      pojemność baza emiter;  
CBC      pojemność baza–kolektor;  
CBX      pojemność baza wewnętrzna–kolektor;  
CJS      pojemność kolektor–podłoże;  
BETAAC   małosygnałowe wzmocnienie prądowe;  
FT        częstotliwość graniczna.

- ☐ Czas pracy jednostki centralnej komputera.

W zbiorze wyjściowym najistotniejsze dla nas są potencjały węzłowe oraz prądy płynące przez tranzystory. Okazuje się, że tranzystory Q1, Q2 oraz Q4 znajdują się w stanie nasycenia, natomiast tranzystor Q3 jest zablokowany.


Jeżeli dane wejściowe uzupełnimy o następujące dwie linie:

```
.DC V2 0 5 0.05  
.PROBE V(8)
```

to w wyniku analizy otrzymamy statyczną charakterystykę przejściową przykładowej bramki TTL. Jak można się zorientować trudności obliczeniowe występują w obszarze przejściowym między stanem, w którym bramka ma duże wzmocnienie, a stanem w którym na wyjściu bramki panuje napięcie bliskie zeru i jednocześnie wzmocnienie bramki wynosi zero. Jest rzeczą charakterystyczną, że problemy ze znalezieniem statycznego punktu pracy występują często w przypadku układów o dużym stałoprądowym wzmocnieniu napięciowym.

### 2.3.3. Parametryzacja źródeł [28]

Powróćmy jeszcze raz do poprzedniego przykładu, w którym obliczany był statyczny punkt pracy bramki TTL (strona 23). W przypadku, gdy nie użyto instrukcji .NODESET (z odpowiednimi parametrami) podczas analizy ekran komputera wyglądał tak jak np. na




**Rys.14.** Charakterystyka statyczna bramki TTL.

Rys. 15. Zwróćmy uwagę na komunikat:

Power supplies cut back to 25%

Oznacza on, że wobec niebezpieżności obliczeń mających na celu znalezienie statycznego punktu pracy układu klasyczną metodą Newton–a Raphson–a zastosowana została specjalna procedura obliczeń. Jej idea jest następująca: w układzie, w którym wszystkie źródła napięcia i prądu mają wydajność równą zero potencjały węzłowe i prądy płynące przez źródła napięcia są równe zero. Jeżeli powiększymy wydajność źródeł do kilku procent ich wydajności nominalnej to należy się spodziewać, że stosując zwykły algorytm Newton–a Raphson–a i zaczynając iteracje od zerowych wartości potencjałów węzłowych łatwo znajdziemy punkt pracy układu. Otrzymany punkt pracy może posłużyć dalej jako punkt początkowy do obliczania punktu pracy po dalszym powiększeniu wydajności źródeł. Jeżeli na każdym etapie tej procedury jesteśmy w stanie znaleźć statyczny punkt pracy to w momencie, gdy źródła osiągną swoją wydajność nominalną obliczony punkt pracy jest właściwym punktem pracy układu. Metoda ta nazywana jest metodą parametryzacji źródeł. W istotny sposób polepsza zbieżność obliczeń statycznego punktu pracy układu i mimo, że w pierwszym naszym przykładzie zawiodła, Czytelnik w toku dalszego użytkowania programu PSpice przekona się, że w wielu przypadkach oddaje nieocenione usługi.



**Rys.15.** Obliczanie statycznego punktu pracy bramki TTL. Wygląd ekranu monitora komputerowego.

#### 2.4. Transmitancje stałoprądowe

Obliczanie statycznego punktu pracy układu i dowolnej charakterystyki stałoprądowej nie wyczerpuje możliwości analizy stałoprądowej programu PSpice. Możliwe jest bowiem obliczenie obliczenie dowolnej małosygnałowej transmitancji stałoprądowej.

Transmitancja małosygnałowa  $T_{xy}$  od wielkości wejściowej  $x$ , którą może być wartość dowolnego wymuszenia w obwodzie, do wielkości wyjściowej  $y$ , którą może być dowolne napięcie lub prąd w obwodzie definiowane jest jako pochodna cząstkowa wyjścia  $y$  względem wejścia  $x$ :

$$T_{xy} = \left. \frac{\partial y}{\partial x} \right|_{(\text{wymuszenia} \neq x) = \text{const.}} \quad (23)$$

W przypadku obwodu liniowego, w którym  $x$  jest jedynym wymuszeniem transmitancja  $T_{xy}$  sprowadza się do stosunku wielkości wyjściowej  $y$  i wielkości wejściowej  $x$ . Wartość transmitancji nie zależy przy tym od wartości wymuszenia  $x$ :

$$T_{xy} = \frac{y}{x} \neq f(x) \quad (24)$$

Transmitancja stałoprądowa oznacza transmitancję obliczoną w obwodzie, w którym każda z pojemności jest traktowana jako rozwarcie, natomiast każda z indukcyjności obwodu jest traktowana jako zwarcie.

#### 2.4.1. Instrukcja .TF — obliczanie transmitancji

Obliczenie transmitancji stałoprądowej zleca się programowi PSpice za pomocą instrukcji .TF (ang. transmitance function — transmitancja). Jej postać jest następująca:

**.TF \_wYjście \_wEjście**

Przykłady:

```
.TF V(1,2) VIN
.TF I(VOOUT) IIN
.TF V(2) ICNTRL
```

Wielkość wyjściowa, określona w polu **\_wYjście**, może być:

- ☐ Dowolne napięcie w obwodzie. Napięcie pomiędzy węzłem o numerze **\_n1** i węzłem o numerze **\_n2** zapisywane jest następująco:

$V\_n1,[_n2])$

Parametr **\_n2** jest opcjonalny. Jeżeli nie podamy go zostanie przyjęte założenie, że **\_n2=0** tzn. drugi węzeł to węzeł masy. W tym przypadku napięcie sprowadza się do potencjału węzłowego.

- ☐ Prąd płynący przez dowolne niezależne źródło napięcia. Prąd taki zapisywany jest następująco:


$I(v\_nazwa)$

Parametr **v\_nazwa** oznacza nazwę SEM, przez którą płynie interesujący nas prąd.

Parametr **\_wEjście** w instrukcji .TF oznacza wielkość sterującą. Może nią być:

- ☐ Wartość napięcia wymuszanego przez dowolne niezależne źródło napięcia (SEM). W polu **\_wEjście** należy w tym przypadku podać nazwę źródła napięcia.

- ☐ Wartość prądu wymuszanego przez dowolne niezależne źródło prądu (SPM). W polu **\_wEjście** należy w tym przypadku podać nazwę źródła prądu.



**Rys.16.** Wzmacniacz różnicowy.

Przykład:

Obliczyć wzmocnienie napięciowe układu wzmacniacza różnicowego z Rys. 16. Wielkością wejściową (sterującą) niech będzie wartość napięcia źródła VIN, natomiast wielkością wyjściową napięcie pomiędzy węzłami o numerze 2 i 5. Należy przyjąć, że tranzystory powstały w jednym procesie technologicznym.

Poszukiwane wzmocnienie napięciowe to transmitancja od wymuszenia VIN do napięcia V(5,2). Aby je obliczyć można posłużyć się instrukcją .TF. Dane dla programu PSpice przedstawione są poniżej.

---

```

WZMACNIACZ ROZNICOWY
.MODEL TRANZYSTOR NPN BF=100           ;model mat. tranzystora npn
VCC 1 0 5V                             ;źródło dodatniego napięcia zasilania
VEE 0 6 5V                             ;źródło ujemnego napięcia zasilania
VIN 3 0                                 ;źródło sterujące
RE 4 6 2.2K                            ;oporność emiterowa
R1 1 2 2.2K                             ;oporność kolektorowa
R2 1 5 2.2K                             ;oporność kolektorowa
Q1 2 3 4 TRANZYSTOR                    ;para tranzystorów
Q2 5 0 4 TRANZYSTOR
.TF V(5,2) VIN                         ;oblicz transmitancję
.END                                    ;koniec danych

```

---

Program PSpice oblicza statyczny punkt pracy układu, a następnie żadaną transmitancję. W zbiorze wyjściowym znajdujemy kolejno: opis obwodu, parametry modelu tranzystora, informacje o statycznym punkcie pracy układu i na koniec potrzebną transmitancję. Interesujący nas fragment tego zbioru pokazany jest poniżej.

---

```

****      SMALL-SIGNAL CHARACTERISTICS

V(5,2)/VIN = 8.091E+01

INPUT RESISTANCE AT VIN = 5.405E+03      <-- Oporność układu widziana z zacisków źródła VIN

OUTPUT RESISTANCE AT V(5,2) = 4.400E+03  <-- Oporność wyjściowa widziana z zacisków 2 i 5

```

---


Obliczone przez program PSpice wzmocnienie napięciowe układu wynosi 80.91[V/V]. Oprócz tego obliczona została:

- ☐ Oporność (dynamiczna) układu widziana z zacisków źródła VIN — 5.405kΩ:
- ☐ Oporność wyjściowa układu widziana z zacisków (węzłów) 2 i 5 — 4.4kΩ.

#### 2.4.2. Inny sposób obliczania transmitancji stałoprądowej

Te same rezultaty można otrzymać postępując inaczej. W tym celu wykorzystać można program graficzny Probe (dodatek D). Program ten nie tylko służy do graficznej prezentacji wyników obliczeń ale także służy do dalszego ich przetwarzania. W szczególności jest w stanie zróżniczkować (numerycznie) dowolną z wyświetlanych charakterystyk. Zatem obliczymy charakterystykę statyczną wzmacniacza różnicowego rozważanego w ostatnim





**Rys.17.** Wzmacniacz różnicowy. a) Napięcie wyjściowe  $V(5,2)$  w funkcji napięcia sterującego VIN. b) Wzmocnienie napięciowe w funkcji wysterowania VIN.

przykładzie. Zmienną niezależną niech będzie wartość siły elektromotorycznej VIN, natomiast wyjście niech stanowi różnica potencjałów między węzłami 5 i 2 —  $V(5,2)$ . W zbiorze danych wejściowych, po opisie struktury wzmacniacza, umieszczamy instrukcję obliczania charakterystyk stałoprądowych:

.DC VIN -100mV 100mV 1mV

Obliczona charakterystyka przekazana zostanie do programu Probe za pomocą instrukcji .PROBE:

.PROBE V(5,2)

Za pomocą programu Probe wykreślamy zależność napięcia  $V(5,2)$  od wartości napięcia wymuszającego VIN Rys. 17a. Można też wyświetlić pochodną napięcia  $V(5,2)$  względem VIN jako funkcję wartości wymuszenia VIN<sup>9</sup>. Wartość tej pochodnej dla  $VIN=0[V]$  to poszukiwane wzmocnienie (transmitancja) układu — Rys. 17b.

<sup>9</sup>Wzmacniacz różnicowy jest nieliniowym układem elektronicznym, w związku z tym jego wzmocnienie jest funkcją poziomu wysterowania - VIN.

## 2.5. Wrażliwości

Obok charakterystyk i transmitancji program PSpice może posłużyć także do obliczania wrażliwości dla obwodów prądu stałego. Obliczane są dwa typy wrażliwości.

- Wrażliwości bezwzględne. Wrażliwością bezwzględną wielkości  $Y(x,z)$  na zmiany parametru  $x$ , przy ustalonej wartości parametru  $z$  nazywamy pochodną cząstkową  $Y(x,z)$  względem  $x$ .

$$Qb_x^Y = \frac{\partial Y(x,z)}{\partial x} \quad (25)$$

- Wrażliwości półwzględne. Wrażliwością półwzględną wielkości  $Y(x,z)$  na zmiany parametru  $x$ , przy ustalonej wartości parametru  $z$  nazywamy iloczyn parametru  $x$  i wrażliwości bezwzględnej wielkości  $Y(x,z)$  na zmiany parametru  $x$ , przy ustalonej wartości parametru  $z$ .

$$Q_x^Y = x \cdot \frac{\partial Y(x,z)}{\partial x} = x \cdot Qb_x^Y \quad (26)$$

Dane wyjściowe zawierają wrażliwości półwzględne mnożone przez czynnik 0.01. Ułatwia to posługiwanie się tolerancjami parametrów układu wyrażonymi w procentach.

### 2.5.1. Format instrukcji

Format instrukcji `.SENS` (ang. sensitivity — wrażliwość) powodującej, że obliczane są wrażliwości stałoprądowe jest następujący:

**`.SENS <l_wYjść>`**

Przykład:

`.SENS V(7) V(7,9) I(VCC)`


Instrukcja `.SENS` powoduje, że obliczone zostaną wrażliwości wielkości umieszczonych na liście `<l_wYjść>` względem zmian parametrów wszystkich elementów w obwodzie. Wielkością, która może znaleźć się na liście `<l_wYjść>` może być potencjał węzłowy (np. `V(7)`), różnica potencjałów węzłowych (np. `V(7,9)`), prąd płynący przez dowolne źródło napięcia (np. `I(VIN)`) lub inna wielkość, której przebieg można wykreślić za pomocą instrukcji `.PRINT` (strona 52). Należy zawsze pamiętać, że w przypadku dużych obwodów instrukcja `.SENS` powoduje umieszczenie w zbiorze wyjściowym dużej liczby danych.

### 2.5.2. Projektowanie przetwornika C/A

Analiza wrażliwościowa może zostać zastosowana wszędzie tam, gdzie konieczne jest zoptymalizowanie pewnych własności obwodu. Przedmiotem optymalizacji może być struktura

obwodu, wartość dowolnie wybranej wielkości w obwodzie (w szczególności transmitancji), koszt wyprodukowania obwodu itp.

Przykład:



**Rys.18.** Drabinka oporników ośmiobitowego przetwornika C/A.

Na Rys. 18 przedstawiona jest drabinka oporowa, która stanowi podstawową część 8 bitowego przetwornika cyfrowo–analogowego (C/A). Dwupołożeniowe klucze sterowane są przez kolejne bity przetwarzanej liczby. Jeżeli cyfra binarna równa jest 0, odpowiedni klucz jest zwarty do masy. Jeśli cyfra binarna równa jest 1, odpowiedni klucz zwarty jest do źródła zasilania. Można wykazać, że napięcie na wyjściu układu (między węzłem oznaczonym na rysunku WY, a masą) wyraża się następującym wzorem:

$$U_{wy} = U_0 * \left( \sum_{i=1}^8 K_i * 2^{-i+1} \right) \quad (27)$$

Liczby K1–K8 przyjmują wartości 0 lub 1 stosownie do położenia odpowiednich kluczy. Należy tak dobrać wartość tolerancji poszczególnych oporników aby wartość błędu napięcia wyjściowego, wynikająca z rozrzutu wartości oporności nie przekraczała 1/2 zmiany napięcia wyjściowego, spowodowanej zmianą na najmniej znaczącej pozycji przetwarzanej liczby (klucz K8). Należy przyjąć, że wartość oporności  $R=10[\text{k}\Omega]$ , natomiast wartość wymuszenia  $V_{ref}=9[\text{V}]$ .

W celu analizy za pomocą programu PSpice rozważany obwód wygodnie jest doprowadzić do postaci przedstawionej na Rys. 19. Siły elektromotoryczne V1–V8 mogą przyjmować wartość 0[V] lub 9[V]. Ponieważ obwód jest liniowy napięcie wyjściowe V(9) jest liniową kombinacją wartości poszczególnych SEM.

$$U_{wy} = \sum_{i=1}^8 a_i \cdot V_i \quad (28)$$

Współczynniki  $a_1$ – $a_8$  są uzależnione tylko od wartości oporności  $R_1$ – $R_{16}$ . Wartości napięć  $V_1$ – $V_8$  zależą natomiast od stanu poszczególnych kluczy (przetwarzanej liczby). Załóżmy, że tolerancje oporności  $\delta R_i$  są niewielkie:

$$\delta R_i = \frac{\Delta R_i}{R_i} \quad (29)$$

gdzie:

- $\delta R_i$  tolerancja opornika  $R_i$ ;
- $\Delta R_i$  bezwzględna wartość odchyłki oporności od wartości nominalnej;
- $R_i$  wartość nominalna oporności.

Jeżeli rozwinąć zależność  $U_{wy}$  w szereg Taylor–a wokół wartości nominalnych oporności i odrzucić wyrazy rzędu wyższego niż 1 (zakładamy małe tolerancje) to błąd napięcia  $\Delta U_{wy}$  można wyrazić w sposób następujący:

$$\Delta U_{wy} \approx \sum_{i=1}^{16} \frac{\partial U_{wy}}{\partial R_i} \cdot \Delta R_i = \sum_{i=1}^{16} Q_{Ri}^{U_{wy}} \cdot \delta R_i \quad (30)$$

Moduł prawej strony powyższego równania powinien być nie większy od dopuszczalnego błędu przetwarzania  $\Delta U_{wy}$ . Prowadzi to do nierówności:

$$\Delta U_{wy} \geq \sum_{i=1}^{16} \left| Q_{Ri}^{U_{wy}} \right| \cdot \delta R_i \quad (31)$$

Aby nierówność powyższa zachodziła dla każdej wartości napięcia wyjściowego to wystarczy aby spełniona była nierówność:

$$\Delta U_{wy} \geq \sum_{i=1}^{16} \left| \max_{U_{wy}} \{ Q_{Ri}^{U_{wy}} \} \right| \cdot \delta R_i \quad (32)$$

Jeżeli przyjmiemy, że wpływ każdego z szesnastu oporników drabinki na błąd przetwarzania powinien być ten sam to wzór na tolerancję poszczególnych oporników jest następujący:

$$\left| \delta R_i \right| \leq \frac{\Delta U_{wy}}{16 \cdot \left| \max_{U_{wy}} \{ Q_{Ri}^{U_{wy}} \} \right|} \quad (33)$$

Pozostaje problem obliczenia, dla każdego opornika osobno, maksymalnej wartości modułu wrażliwości napięcia wyjściowego względem jego oporności. Do tego celu można

wykorzystać program PSpice. Zgodnie ze wzorem (28) odpowiednie wrażliwości wyrażają się wzorem:

$$Q_{Rk}^{U_{wy}} = \sum_{i=0}^8 Q_{Rk}^{a_i} \cdot V_i \quad (34)$$

Składnik o numerze i sumy występującej po prawej stronie powyższego wzoru jest równy:

- ☐ Zero, jeśli wartość siły elektromotorycznej  $V_i=0[V]$ .
- ☐ Wrażliwości napięcia wyjściowego na zmiany oporności  $R_k$  dla  $V_i=9[V]$ ;  $V_j=0[V]$  przy  $j \neq i$ .

Wynika stąd, że za pomocą programu PSpice należy obliczyć wrażliwości napięcia wyjściowego na zmiany wszystkich oporników w następujących ośmiu przypadkach:

- ☐ **V1=9[V]**,  $V_2=0[V]$ ,  $V_3=0[V]$ ,  $V_4=0[V]$ ,  $V_5=0[V]$ ,  $V_6=0[V]$ ,  $V_7=0[V]$ ,  $V_8=0[V]$ .
- ☐  $V_1=0[V]$ , **V2=9[V]**,  $V_3=0[V]$ ,  $V_4=0[V]$ ,  $V_5=0[V]$ ,  $V_6=0[V]$ ,  $V_7=0[V]$ ,  $V_8=0[V]$ .
- ☐  $V_1=0[V]$ ,  $V_2=0[V]$ , **V3=9[V]**,  $V_4=0[V]$ ,  $V_5=0[V]$ ,  $V_6=0[V]$ ,  $V_7=0[V]$ ,  $V_8=0[V]$ .
- ☐  $V_1=0[V]$ ,  $V_2=0[V]$ ,  $V_3=0[V]$ , **V4=9[V]**,  $V_5=0[V]$ ,  $V_6=0[V]$ ,  $V_7=0[V]$ ,  $V_8=0[V]$ .
- ☐  $V_1=0[V]$ ,  $V_2=0[V]$ ,  $V_3=0[V]$ ,  $V_4=0[V]$ , **V5=9[V]**,  $V_6=0[V]$ ,  $V_7=0[V]$ ,  $V_8=0[V]$ .
- ☐  $V_1=0[V]$ ,  $V_2=0[V]$ ,  $V_3=0[V]$ ,  $V_4=0[V]$ ,  $V_5=0[V]$ , **V6=9[V]**,  $V_7=0[V]$ ,  $V_8=0[V]$ .
- ☐  $V_1=0[V]$ ,  $V_2=0[V]$ ,  $V_3=0[V]$ ,  $V_4=0[V]$ ,  $V_5=0[V]$ ,  $V_6=0[V]$ , **V7=9[V]**,  $V_8=0[V]$ .
- ☐  $V_1=0[V]$ ,  $V_2=0[V]$ ,  $V_3=0[V]$ ,  $V_4=0[V]$ ,  $V_5=0[V]$ ,  $V_6=0[V]$ ,  $V_7=0[V]$ , **V8=9[V]**.

Następnie dla każdego opornika należy rozważyć wszystkie możliwe kombinacje wartości sił elektromotorycznych  $V_i$  i wybrać tę która daje największą wartość modułu wrażliwości. Ponieważ dana SEM może mieć tylko wartość  $0[V]$  (brak składnika sumy we wzorze (34)) lub  $9[V]$  (składnik jest uwzględniany) to wystarczy zsumować:

- ☐ wrażliwości o znaku dodatnim;
- ☐ wrażliwości o znaku ujemnym.


Z utworzonych w ten sposób sum wybieramy tę, której wartość bezwzględna jest większa. Wartość bezwzględna wybranej sumy to maksymalny moduł wrażliwości napięcia wyjściowego na zmiany danego opornika.

Poniżej przedstawione są przykładowe dane dla programu PSpice. Siła elektromotoryczna  $V_1$  przyjmuje wartość  $9V$  natomiast wszystkie pozostałe SEM są równe zeru. Jedyną instrukcją dla programu PSpice jest instrukcja .SENS polecająca obliczenie wrażliwości potencjału węzłowego  $V(9)$  na zmiany parametrów wszystkich elementów w obwodzie.

PRZETWORNIK C/A

1

```
V1 1 0 9V
V2 2 0
V3 3 0
V4 4 0
V5 5 0
V6 6 0
V7 7 0
V8 8 0
R1 9 0 20K
R2 9 1 20K
R3 10 2 20K
R4 11 3 20K
R5 12 4 20K
R6 13 5 20K
R7 14 6 20K
R8 15 7 20K
R9 16 8 10K
R10 9 10 10K
R11 10 11 10K
R12 11 12 10K
R13 12 13 10K
R14 13 14 10K
R15 14 15 10K
R16 15 16 10K
.SENS V(9)
.END
```



**Rys.19.** Drabinka oporników ośmiobitowego przetwornika C/A. Obwód analizowany za pomocą programu PSpice.

W zbiorze z danymi wyjściowymi, oprócz wartości wrażliwości, znajdujemy obliczone przez program PSpice, potencjały węzłowe. Załóżmy, że przetwarzana liczba zmienia się na najmniej znaczącej pozycji. Zmiana napięcia wyjściowego równa jest potencjałowi węzłowemu V(9) (napięcie wyjściowe) w przypadku, gdy klucz K8 jest zwarty do napięcia odniesienia zaś reszta kluczy zwarta jest do masy. Dopuszczany błąd przetwarzania  $\Delta U_{wy}$  jest równy połowie wspomnianego napięcia. Zatem, zgodnie z obliczeniami, dopuszczalny błąd  $\Delta U_{wy}=23.5[\text{mV}]^{10}$ . Tablica II zawiera obliczone przez program PSpice wartości wrażliwości wraz z wartościami tolerancji, obliczonymi zgodnie ze wzorem (33). W ostatniej kolumnie podano wartości tolerancji poszczególnych oporników zaokrąglone do najbliższej wartości w szeregu tolerancji.

Zwróćmy uwagę na następujące fakty, dotyczące modułów współczynników wrażliwości:

- ☐ Im dalej od wyjścia położony jest klucz zwarty do źródła napięcia odniesienia, tym mniejszy jest wpływ każdego z oporników na napięcie wyjściowe.
- ☐ W sytuacji, gdy do napięcia odniesienia zwarty jest tylko jeden klucz, im dalej od wyjścia położony jest opornik w strukturze obwodu, tym mniejszy wpływ na napięcie wyjściowe ma wartość jego oporności.

### 2.5.3. Analiza Monte Carlo

Przed przystąpieniem do wytwarzania zaprojektowanej drabinki oporowej przetwornika

<sup>10</sup>Wartość dopuszczanego przez nas błędu można oczywiście obliczyć "ręcznie". Wynosi ona  $1/3 \cdot U_{ref} \cdot 2^{-7} = 23.4375[\text{mV}]$ .

**Tablica II** Wrażliwości drabinki oporowej przetwornika C/A obliczone przez program PSpice.

Nazwa	Maksymalny moduł sumy wrażliwości										
Wrażliwości półwzględne (wołty/procenty)								Tolerancje[%]			
	V1=9V	V2=9V	V3=9V	V4=9V	V5=9V	V6=9V	V7=9V	V8=9V	Tolerancje zaokrąglone[%]		
R1	1.000E-2	5.000E-3	2.500E-3	1.250E-3	6.250E-4	3.125E-4	1.563E-4	1.563E-4	2.000E-2	7.324E-2	0.05
R2	-2.000E-2	5.000E-3	2.500E-3	1.250E-3	6.250E-4	3.125E-4	1.563E-4	1.563E-4	2.000E-2	7.324E-2	0.05
R3	2.500E-3	-1.000E-2	2.500E-3	1.250E-3	6.250E-4	3.125E-4	1.563E-4	1.563E-4	1.000E-2	1.465E-1	0.10
R4	6.250E-4	1.250E-3	-5.000E-3	1.250E-3	6.250E-4	3.125E-4	1.563E-4	1.563E-4	5.000E-3	2.930E-1	0.20
R5	1.563E-4	3.125E-4	6.250E-4	-2.500E-3	6.250E-4	3.125E-4	1.563E-4	1.563E-4	2.500E-3	5.859E-1	0.50
R6	3.906E-5	7.813E-5	1.563E-4	3.125E-4	-1.250E-3	3.125E-4	1.563E-4	1.563E-4	1.250E-3	1.172E+0	1.00
R7	9.766E-6	1.953E-5	3.906E-5	7.813E-5	1.563E-4	-6.250E-4	1.563E-4	1.563E-4	6.250E-4	2.344E+0	2.00
R8	2.441E-6	4.883E-6	9.766E-6	1.953E-5	3.906E-5	7.813E-5	-3.125E-4	1.563E-4	3.125E-4	4.688E+0	2.00
R9	1.221E-6	2.441E-6	4.883E-6	9.766E-6	1.953E-5	3.906E-5	7.813E-5	-1.563E-4	1.563E-4	9.372E+0	5.00
R10	5.000E-3	-5.000E-3	-2.500E-3	-1.250E-3	-6.250E-4	-3.125E-4	-1.563E-4	-1.563E-4	1.000E-2	1.465E-1	0.10
R11	1.250E-3	2.500E-3	-2.500E-3	-1.250E-3	-6.250E-4	-3.125E-4	-1.563E-4	-1.563E-4	5.000E-3	2.930E-1	0.20
R12	3.125E-4	6.250E-4	1.250E-3	-1.250E-3	-6.250E-4	-3.125E-4	-1.563E-4	-1.563E-4	2.500E-3	5.859E-1	0.50
R13	7.813E-5	1.563E-4	3.125E-4	6.250E-4	-6.250E-4	-3.125E-4	-1.563E-4	-1.563E-4	1.250E-3	1.172E+0	1.00
R14	1.953E-5	3.906E-5	7.813E-5	1.563E-4	3.125E-4	-3.125E-4	-1.563E-4	-1.563E-4	6.251E-4	2.343E+0	2.00
R15	4.883E-6	9.766E-6	1.953E-5	3.906E-5	7.813E-5	1.563E-4	-1.563E-4	-1.563E-4	3.126E-4	4.686E+0	2.00
R16	1.221E-6	2.441E-6	4.883E-6	9.766E-6	1.953E-5	3.906E-5	7.813E-5	-1.563E-4	1.563E-4	9.372E+0	5.00

C/A warto sprawdzić poprawność projektu. Bliższa analiza danych dotyczących wrażliwości i tolerancji oporników<sup>11</sup> wskazuje, że maksymalny błąd napięcia wyjściowego wywołany niedokładnością wykonania oporników występuje podczas przetwarzania liczby 169. Odpowiada to sytuacji, w której klucze K1, K3, K5, K8 są zwarte do źródła napięcia odniesienia Vref natomiast pozostałe klucze zwarte są do masy. Dla takiej kombinacji kluczy wykonamy, za pomocą programu PSpice, analizę Monte Carlo. Będzie ona polegała na tym, że program PSpice wylosuje automatycznie wartości oporników, tak by mieściły się w podanym zakresie tolerancji. Następnie dokonana zostanie analiza układu. W zbiorze wyjściowym znajdzie się wartość odchylenia napięcia wyjściowego od wartości nominalnej. Losowanie wartości elementów powtarzane jest przez program PSpice zadaną liczbę razy<sup>12</sup>. Analiza Monte Carlo zlecona jest programowi PSpice za pomocą następującej instrukcji.

**.MC \_liczba \_rodzaj\_analizy \_wyjście [YMAX] [LIST] [OUTPUT \_typ]**

Przykłady:

```
.MC 50 DC IC(Q7) YMAX LIST
.MC 100 AC V(5) OUTPUT EVERY 10
.MC 10 TRAN V(5) YMAX
```

<sup>11</sup>Napisany został program w C, który na podstawie obliczonych przez program PSpice wrażliwości i dobranych tolerancji oporników przeanalizował błędy napięcia wyjściowego powstające podczas przetwarzania 256 możliwych liczb.

<sup>12</sup>W praktyce ograniczeniem jest wielkość pamięci operacyjnej potrzebna programowi do posortowania wyników.

Typ analizy, która będzie powtarzana, określa parametr **\_rodzaj\_analizy**. Może on przyjąć trzy wartości:

DC	obliczanie charakterystyk statycznych;
AC	małosygnałowa analiza zmiennoprądowa;
TRAN	analiza stanów nieustalonych.

Należy pamiętać, że w zbiorze danych wejściowych *musi* znaleźć się instrukcja zlecająca wykonanie analizy, której dotyczy analiza Monte Carlo. Liczbę powtórzeń określa parametr **\_liczba**. Analiza Monte Carlo dotyczyć będzie wielkości wyjściowej o nazwie określonej przez parametr **\_wyjście**. Może być nim dowolny potencjał węzłowy, różnica potencjałów węzłowych, prąd płynący przez dowolne niezależne źródło napięcia lub inna wielkość, tak jak dopuszcza to instrukcja **.PRINT** (patrz strona 52). Użycie słowa kluczowego **YMAX** powoduje, że w zbiorze wyjściowym umieszczone zostanie tylko największe odchylenie wielkości wyjściowej od jej wartości nominalnej. Słowo kluczowe **LIST** powoduje umieszczenie w zbiorze wyjściowym, podczas każdej z analiz, aktualnych wartości zmienianych parametrów obwodu. Słowo kluczowe **OUTPUT** wraz z parametrem **\_typ** powoduje, że oprócz wyników pierwszej analizy w zbiorze wyjściowym umieszczone zostaną wyniki kolejnych analiz tak jak to określa parametr **\_typ**. Parametr **\_typ** przyjmować może następujące wartości:

- ALL** — w zbiorze wyjściowym umieszczone zostaną wyniki wszystkich analiz;
- FIRST n** — w zbiorze wyjściowym umieszczone zostaną wyniki pierwszych n analiz;
- EVERY n** — w zbiorze wyjściowym umieszczone zostaną wyniki co n-tej analizy;
- RUN <\_lista>** — w zbiorze wyjściowym umieszczone zostaną wyniki analiz o numerach wyszczególnionych na liście <\_lista>.

Tolerancje wartości elementów mogą być określone tylko za pomocą deklaracji modelu **.MODEL** — po każdym parametrze modelu może wystąpić słowo kluczowe **DEV** lub **LOT**.

**DEV \_tolerancja [%]** — deklaruje, że parametr posiada tolerancję o wartości określonej przez parametr **\_tolerancja**. Jeśli wystąpi za nim znak „, % ” to wartość tolerancji podana jest w procentach. Wartość parametru *zmienia się niezależnie* dla wszystkich elementów odwołujących się do modelu.

**LOT \_tolerancja [%]** — deklaruje, że parametr posiada tolerancję o wartości określonej przez parametr **\_tolerancja**. Jeśli wystąpi za nim znak „, % ” to wartość tolerancji podana jest w procentach. Wartość parametru jest *taka sama* dla wszystkich elementów odwołujących się do modelu.

W przypadku naszego przetwornika C/A dane dla programu PSpice, potrzebne dla wykonania analizy Monte Carlo przedstawione są poniżej.

---

```
PRZETWORNIK C/A - ANALIZA MONTE CARLO
V1 1 0 9V
V2 2 0
```



---

```

V3 3 0 9V
V4 4 0
V5 5 0 9V
V6 6 0
V7 7 0
V8 8 0 9V
.MODEL R005 RES R=1 DEV=0.05%
.MODEL R01 RES R=1 DEV=0.1%
.MODEL R02 RES R=1 DEV=0.2%
.MODEL R05 RES R=1 DEV=0.5%
.MODEL R10 RES R=1 DEV=1%
.MODEL R20 RES R=1 DEV=2%
.MODEL R50 RES R=1 DEV=5%
R1 9 0 R005 20K
R2 9 1 R005 20K
R3 10 2 R01 20K
R4 11 3 R02 20K
R5 12 4 R05 20K
R6 13 5 R10 20K
R7 14 6 R20 20K
R8 15 7 R20 20K

R9 16 8 R50 10K

R10 9 10 R01 10K
R11 10 11 R02 10K
R12 11 12 R05 10K
R13 12 13 R10 10K
R14 13 14 R20 10K
R15 14 15 R20 10K
R16 15 16 R50 10K

.DC V1 9 9 1
.MC 600 DC V(9) YMAX
.END

```

---


Zdefiniowano tam siedem modeli oporników<sup>13</sup>. Są to oporniki o tolerancji 0.05%, 0.1%, 0.2%, 0.5%, 1.0%, 2.0%, 5.0%. W deklaracji .MODEL nie podaje się wartości opornika a jedynie liczbę, przez którą zostanie przemnożona wartość oporności podana w linii deklaracji opornika. Należy przy tym zauważyć, że w deklaracji opornika, która odwołuje się do modelu opornika, wartość oporności poprzedzona jest nazwą modelu. Ponieważ analiza Monte Carlo może dotyczyć tylko analizy DC, AC lub TRAN wykonano analizę DC, podczas której wartość napięcia źródła V1 zmieniała się od 9[V] do 9[V] z krokiem 1[V], co jest równoważne obliczeniu statycznego punktu pracy układu. Analiza Monte Carlo dotyczyła potencjału  $V(9)=U_{wy}$  przy czym liczba powtórzeń wynosiła 600. Dziesięć największych odchyleń napięcia wyjściowego od wartości nominalnej, obliczonych przez program PSpice i umieszczonych w zbiorze z wynikami, podanych jest poniżej.

W sześciuset próbach nie zauważono błędu napięcia wyjściowego większego niż 4.455[mV], chociaż zgodnie z obliczeniami teoretycznymi największy możliwy błąd wynosi około 8.9[mV]. Wyniki otrzymane podczas analizy Monte Carlo były dalej przetwarzane. Na Rys. 20 przedstawiono uzyskany na tej podstawie histogram modułu błędu napięcia wyjściowego w przypadku, gdy przetwarzana była liczba 169. Stanowi on pewnego rodzaju

---

<sup>13</sup>Bliższe omówienie modelu opornika znajduje się w rozdziale 5.

estymatę funkcji gęstości prawdopodobieństwa błędu przetwarzania. Widać z niego, że błędy przetwarzania wynikające z niedoskonałości oporników praktycznie nie przekraczają wartości 4.5[mV]. Wynika to stąd, że kombinacje wartości oporności powodujące powstawanie maksymalnych błędów przetwarzania zdarzają się bardzo rzadko. Jeśli uwzględnić ten fakt i dokonać analizy statystycznej naszego układu okaże się, że wszystkie tolerancje mogą zostać powiększone aż czterokrotnie! Mimo to błędy przetwarzania wynikające z tolerancji oporników większe niż  $4 \cdot 4.5[\text{mV}] = 18[\text{mV}]$  będą niezwykle rzadkie.



**Rys.20.** Histogram wyników analizy Monte Carlo. Na osi poziomej odłożono moduł błędu przetwarzania.

\*\*\*\*\* 08/14/91 \*\*\*\*\* Evaluation PSpice (Jan. 1988) \*\*\*\*\* 10:31:06 \*\*\*\*\*

PRZETWORNIK C/A - ANALIZA MONTE CARLO

\*\*\*\* SORTED DEVIATIONS OF V(9) TEMPERATURE = 27.000 DEG C

MONTE CARLO ANALYSIS SUMMARY

\*\*\*\*\*

RUN	MAX DEVIATION	
177	4.455E-03	> than nominal at V1 = 9
215	4.294E-03	< than nominal at V1 = 9
174	3.960E-03	> than nominal at V1 = 9
369	3.880E-03	> than nominal at V1 = 9
546	3.817E-03	> than nominal at V1 = 9
113	3.425E-03	< than nominal at V1 = 9
242	3.332E-03	< than nominal at V1 = 9
467	3.297E-03	< than nominal at V1 = 9
322	3.242E-03	> than nominal at V1 = 9
337	3.178E-03	> than nominal at V1 = 9

Przedstawiony przykład powinien przekonać nas, że za pomocą symulatora można uzyskać bardzo wiele informacji na temat projektowanego układu bez badań prototypowych. Możliwe jest nawet przewidywanie statystycznych parametrów procesu produkcyjnego takich jak np. uzysk produkcyjny. W przypadku układów scalonych jest to jedyna, ekonomicznie rozsądna, droga uzyskania istotnych informacji na temat własności przygotowywanego układu, jeszcze przed rozpoczęciem produkcji.